
SAS syntax and output for Chapter 9: Structural Equation Modeling I: Path Analysis
Much of the material in this document builds on the SAS syntax and output from earlier chapters. If necessary, refer back to those documents for reminders.


Begin by importing the 'lennox.dat' file (for the social support and depression example) into SAS:

	data ch9data;
 infile 'c:\lennox.dat' firstobs=2 dlm='';
 input id cesd pass size density homophily;
 drop id;
run;





As always, one should graph the data and inspect descriptive stats before fitting models:

	proc sgscatter data=ch9data;
 matrix cesd pass size density homophily / diagonal=(kernel);
run;

proc corr data=ch9data noprob;
 var cesd pass size density homophily;
run;




proc means data=ch9data maxdec=2 n mean std median min max range skew kurt;
run;





The scatterplots reproduce Figure 9.10, indicating that the variables are weakly to moderately associated; the correlations among the variables (reported in Table 9.1) are consistent with this observation.

The univariate distributions are approximately normal (univariate skewness and kurtosis values are relatively low). The 'cesd' and 'pass' variables have notably different scales (much larger standard deviations) than the other variables.



Model specification

These examples use the 'calis' procedure to estimate SEMs. 
To estimate a model with ‘proc calis’, the model first needs to be specified under ‘path’.
Here is how to specify the path model in Figure 9.7 for subsequent estimation using ‘proc calis’:

	proc calis data=ch9data;
 path
 cesd <--- pass,
 pass <--- size density homophily;
run;



In the text under ‘path’, two regression equations (one for each endogenous variable in the model) are specified. Each regression equation gets its own line.

Thus, the syntax above indicates that the 'cesd' variable is regressed on the 'pass' variable, while in another equation 'pass' is regressed on 'size', 'density', and 'homophily'.

It is not necessary to specify explicitly the covariances among the exogenous variables or the endogenous variable error variances; ‘proc calis’ will include these parameters by default.

Other model specifications

Specification of the model in Figure 9.1:

	proc calis data=ch9data;
 path
 cesd <--- pass;
run;



Figure 9.2 model:

	proc calis data=ch9data;
 path
pass <--- size density homophily;
run;



Figure 9.3 model:

	proc calis data=ch9data;
 path
 Y1 <--- X1 X2,
 Y2 <--- X1 X2;
 cov
 Y1 Y2 = 0;
run;



The last line above under ‘cov’ specifies that the error covariance parameter between the two endogenous variables, Y1 and Y2, is fixed to zero.

For this particular model, ‘proc calis’ wants to free the error covariance parameter; the last line above prevents that parameter from being free by constraining it to 0 explicitly.

Thus, to sepcify the model of Figure 9.4, only the two regression equations need to be written, thereby accepting the default to allow the error covariance to be free:

	proc calis data=ch9data;
 path
 Y1 <--- X1 X2,
 Y2 <--- X1 X2;
run;



Always check the output of the estimated model to make sure that the model contains the correct error covariance parameters and does not contain any freed covariances which were intended to be fixed to 0.

Figure 9.5 model, simple indirect effect/mediation model:

	proc calis data=ch9data;
 path
 Y1 <--- X,
 Y2 <--- X Y1;
 run;



Figure 9.6 model:

	proc calis data=ch9data;
 path
 Y1 <--- X,
 Y2 <--- X;
 Y3 <--- X Y1 Y2;
run;




Model estimation and fit assessment

Next, running ‘proc calis’ will estimate the model using the default maximum likelihood estimator:

	proc calis data=ch9data;
 path
 cesd <--- pass,
 pass <--- size density homophily;
run;




Consistent with the results reported in Chapter 9, the output above indicates that the model does not fit well according to RMSEA and CFI, although the SRMR value is not bad. The “Bentler-Bonett Non-normed Index” is the same thing as TLI.



Model revision

The standardized residual covariances, in the Z metric, of the model estimated above can be obtained by adding the ‘residual’ option:

	proc calis data=ch9data residual;
 path
 cesd <--- pass,
 pass <--- size density homophily;
run;





Residual correlations can be obtained with

	proc calis data=ch9data corr;
 path
 cesd <--- pass,
 pass <--- size density homophily;
run;





The results above are consistent with those in Table 9.2.
Modificiation indices can be obtained with

	proc calis data=ch9data mod;
 path
 cesd <--- pass,
 pass <--- size density homophily;
run;






Revised model

The syntax below specifies and estimates the revised social support and depression model shown in Figure 9.12:

	proc calis data=ch9data;
 path
 cesd <--- pass density,
 pass <--- size density homophily;
run;

In the output above, RMSEA, CFI, Non-normed Index (i.e., TLI), and SRMR all indicate excellent model fit.
The parameter estimates, standard errors, confidence intervals, and Z tests match the results in Table 9.3.

It's important to understand that the estimates under "Variance Parameters" represent residual variances for the endogenous variables 'cesd' and 'pass'.

GLS estimation

A model can be estimated using GLS instead of the default ML esitmator with the 'method=GLS' option:

	proc calis data=ch9data method=GLS;
 path
 cesd <--- pass density,
 pass <--- size density homophily;
run;

The output above is extremely similar, but not identical, to that obtained with the default ML estimation.


Model comparison

As explained in Chapter 9, the two path models estimated above are formally nested.
Thus, a chi-square difference test (i.e., likelihood ratio test) can be used to determine whether their fit significantly differs:

	data temp;
 chistat = 19.23 - 1.25;
 cdf = 3 - 2;
 pvalue = 1 - probchi(chistat, cdf);
run;

proc print data=temp;
run;





The output above gives the chi-square difference test values reported in the chapter.


Indirect effects

To obtain indirect effect estimates, it is necessary to specify the model so that the indirect effects are explicitly named within the model.

Below, the revised social support and depression model is re-specified to name the indirect effects explicitly using the ‘effpart’ option:

	proc calis data=ch9data;
 path
 cesd <--- pass density,
 pass <--- size density homophily;
 effpart size pass;
 effpart density pass;
 effpart homophily pass;
run;

The indirect effect estimates are the same as those reported in Table 9.4, but bootstrapped confidence intervals for them are not provided.


Assumptions and diagnostics

As explained in the chapter, the example data were simulated from a multivariate normal distribution with linear associations according to a population model specification. Thus, all assumptions are known to be met.

Consequently, the scatterplot matrix and descriptive statistics produced at the beginning of this document illusatrate what data look like under this known population model.
A macro (http://support.sas.com/kb/24/983.html) is available to test for multivariate normality.

